Controlled Natural Language Processing as Answer Set Programming: an Experiment

Rolf Schwitter
Department of Computing
Macquarie University
Sydney NSW 2109, Australia
Rolf.Schwitter@mq.edu.au

Abstract. Most controlled natural languages (CNLs) are processed with the help of a pipeline architecture that relies on different software components. We investigate in this paper in an experimental way how well answer set programming (ASP) is suited as a unifying framework for parsing a CNL, deriving a formal representation for the resulting syntax trees, and for reasoning with that representation. We start from a list of input tokens in ASP notation and show how this input can be transformed into a syntax tree using an ASP grammar and then into reified ASP rules in form of a set of facts. These facts are then processed by an ASP meta-interpreter that allows us to infer new knowledge.

Keywords: Answer Set Programming, Controlled Natural Language Processing, Meta-programming

1 Introduction

Controlled natural languages (CNLs) are subsets of natural languages whose grammars and vocabularies have been restricted in order to eliminate ambiguity and complexity of natural languages for automated reasoning [11, 17]. These CNLs are engineered for a specific purpose and look seemingly informal like natural languages, but they have by design the same properties as their formal target languages. Typically, the writing process of a CNL is supported by an authoring tool that guides the writing of a text or a question by a feedback mechanism [4, 10, 14, 16].

Most existing CNLs are processed with the help of a pipeline architecture that relies on different software components for parsing and translating the CNL input into a formal representation before this representation can be processed by an automated reasoning service [25]. In this paper, we investigate in an experimental way whether answer set programming (ASP) can be used as a unifying framework for CNL processing, knowledge representation and automated reasoning. After a brief introduction to the ASP paradigm in Section 2, we show in Section 3 how grammar rules for a CNL can be written in ASP and how CNL sentences can be parsed into a syntax tree. In Section 4, we discuss how a formal representation can be generated for these syntax trees. In Section 5, we
illustrate how this representation can be used for reasoning in ASP with the help of a meta-interpreter. In Section 6, we summarise our findings and conclude.

2 Answer Set Programming (ASP)

ASP is a form of declarative programming that has its roots in logic programming, disjunctive databases and non-monotonic reasoning [11,13]. ASP provides an expressive formal language for knowledge representation and automated reasoning and is based on the answer set semantics for logic programs [7,8]. In ASP, problems are represented in terms of finite logic theories and these problems are solved by reducing them to finding answer sets which declaratively describe the solutions to these problems. An ASP program consists of a set of rules of the form:

1. \( h_1; \ldots; h_m \leftarrow b_1, \ldots, b_n, \text{not } b_{n+1}, \ldots, \text{not } b_o. \)

where \( h_i \) and \( b_i \) are classical literals. A classical literal is either an atom \( a \) or a negated atom \( \neg a \). A literal of the form \( \neg l \) is a negation as failure literal. The disjunction (\( ; \)) is interpreted as epistemic disjunction [9]. The part on the left of the implication (\( \leftarrow \)) is the head of the rule and the part on the right is the body of the rule. If the body is empty (\( o=0 \)), then we omit the symbol for the implication and end up with a fact. If the head is empty (\( m=0 \)), then we keep the symbol for the implication and end up with an integrity constraint. Note that ASP distinguishes between strong negation (\( \neg \)) and weak negation (\( \text{not} \)); these two forms of negation build the prerequisites for non-monotonic reasoning [9]. For example, the ASP program in (2) consists of two rules, six facts and one integrity constraint:

2. successful(X) :- student(X), work(X), not absent(X).
   -work(X) :- student(X), not work(X).
   student(john). work(john).
   student(sue). work(sue).
   student(mary_ann). absent(mary_ann).
   :- student(X), cheat(X), successful(X).

This program can be processed by an ASP tool such as clingo [6] that computes the following answer set:

3. \{ student(john) work(john) student(sue) work(sue) student(mary_ann)
           absent(mary_ann) successful(sue) successful(john) -work(mary_ann) \}

We call an ASP program satisfiable, if it has at least one answer set. Through inspection of the above answer set, we can immediately see that John and Sue are successful and that Mary Ann does not work. Note that the second rule in (2) specifies the closed world assumption [15] for the literal work/1. If we add the following facts to our program:

4. student(ray). work(ray).

then we end up with an unsatisfiable program since the situation in (4) is excluded by the constraint in (2).
3 Writing a CNL Grammar in ASP

The CNL that we will use in the following discussion is similar to Processable English (PENG) [15] and to Attempto Controlled English (ACE) [5], but the language is less expressive since ASP does not support full first-order logic (FOL). However, ASP is still expressive enough to represent function-free FOL formulas of the $\exists^\ast \forall^\ast$ prefix class in form of a logic program [12]. The following text (5) is written in CNL and expresses the same information as the ASP program in (2):

5. Every student who works and who is not provably absent is successful.
   If a student does not provably work then the student does not work.
   John is a student who works.
   Sue is a student and works.
   Mary Ann who is a student is absent.
   Exclude that a student who cheats is successful.

In order to process this text in ASP, we split it into a sequence of sentences and each sentence into a sequence of tokens. Each token is represented in ASP as a fact (token/4) with four arguments: the first argument holds the string, the second argument holds the sentence number, and the third and fourth argument represent the start and the end position of the string, for example:

6. token("Every", 1, 1, 2). token("student", 1, 2, 3). ...

Each string is stored as a fact (lexicon/5) in the ASP lexicon that distinguishes between function words and content words. Function words (e.g., and, every, who) define the structure of the CNL and content words (e.g., student, works, successful) are used to express the domain knowledge. These lexical entries contain information about the category, the string, the base form, as well as syntactic and semantic constraints (n stands for nil):

7. lexicon(cnj, "and", n, n, n).
   lexicon(det, "Every", n, sg, forall).
   lexicon(rp, "who", n, n, n).
   lexicon(noun, "student", student, sg, n).
   lexicon(iv, "works", work, sg, n).
   lexicon(adj, "successful", successful, n, n).

We can write the grammar for the CNL directly as a set of ASP rules that generate a syntax tree bottom-up, starting from the tokens up to the root. Let us have a look at the grammar rules that process the first sentence in (5). This sentence is interesting, since it contains a coordinated relative clause that is embedded in the main sentence. The first relative clause who works is positive, and the second relative clause who is not provably absent contains a weak negation. It is important to note that this form of negation can only occur in a universally quantified CNL sentence or in a CNL sentence that results in an integrity constraint.

The grammar rule (rules/7) specifies in a declarative way that a sentence (s) starts at position P1 and ends at position P4, if there is a noun phrase (np) that starts at P1 and ends at P2, followed by a verb phrase (vp) that starts at P2 and ends at P3, followed by a punctuation mark (pm) between position P3 and P4:
8. rule(s, s(T1, T2, T3), n, n, N, P1, P4) :-
   rule(np, T1, Y, n, N, P1, P2),
   rule(vp, T2, Y, n, N, P2, P3),
   rule(pm, T3, n, n, N, P3, P4).

The second argument position of this rule is used to build up a syntax tree, the third argument position is used for syntactic constraints, the fourth for semantic constraints, and the fifth for the sentence number. The variable $Y$ in (8) is used to enforce number agreement between the $np$ and the $vp$.

The following grammar rule in (9) further describes the noun phrase of our example sentence. This noun phrase ($np$) consists of a determiner ($det$), followed by a nominal expression ($n1$). The variable $M$ holds a quantifier that controls – as we will see later – the use of weak negation in our example sentence:

9. rule(np, np(T1, T2), Y, n, N, P1, P3) :-
   rule(det, T1, Y, M, N, P1, P2),
   rule(n1, T2, Y, M, N, P2, P3).

The nominal expression ($n1$) expands in our case into a noun ($noun$) and a relative clause ($rcl$):

10. rule(n1, n1(T1, T2), Y, M, N, P1, P3) :-
    rule(noun, T1, Y, n, N, P1, P2),
    rule(rcl, T2, Y, M, N, P2, P3).

The noun ($noun$) is a preterminal category and processes the input token ($token/4$) with the help of the lexical information ($lexicon/5$):

11. rule(noun, noun(S), Y, n, N, P1, P2) :-
    token(S, N, P1, P2),
    lexicon(noun, S, B, Y, n).

Note that the relative clause in our example sentence is coordinated and consists of a positive and a negative part. The grammar rule in (12) for relative clauses ($rcl$) deals with this coordinated structure. In contrast to the positive part, we use the variable $M$ in the negative part of the coordinated structure to enforce that this form of negation occurs under universal quantification (additional grammar rules exist that deal with relative clauses where the order of the positive and negative part is different):

12. rule(rcl, rcl(T1, T2, T3), Y, M, N, P1, P4) :-
    rule(rcl, T1, Y, n, N, P1, P2),
    rule(cnj, T2, n, n, N, P2, P3),
    rule(rcl, T3, Y, M, N, P3, P4).

As the following two grammar rules in (13) illustrate, the relative clause expands in both cases into a relative pronoun ($rp$) followed by a verb phrase ($vp$): the first $vp$ occurs without a variable ($n$) at the fourth argument position and the second $vp$ occurs with a variable ($M$) that holds the quantifier:

13. rule(rcl, rcl(T1, T2), Y, n, N, P1, P3) :-
    rule(rp, T1, n, n, N, P1, P2),
    rule(vp, T2, Y, n, N, P2, P3).
rule(rcl, rcl(T1, T2), Y, M, N, P1, P3).
rule(rp, T1, n, n, N, P1, P2),
rule(vp, T2, Y, M, N, P2, P3).

The first verb phrase (vp) in (13) expands into an intransitive verb (iv):

14. rule(vp, vp(T1), Y, n, N, P1, P2).
rule(iv, T1, Y, n, N, P1, P2).

and the second verb phrase (vp) expands into a copula (cop), followed by a weak negation (naf) and an adjective (adj):

15. rule(vp, vp(T1, T2, T3), Y, M, N, P1, P4).
rule(cop, T1, Y, n, N, P1, P2),
rule(naf, T2, n, M, N, P2, P3),
rule(adj, T3, n, n, N, P3, P4).

As we have seen, weak negation is expressed on the surface level of the CNL with the help of the key phrase *not provably*. The rule (naf) in (16) processes this key phrase if it occurs in the scope of a universal quantifier (forall):

16. rule(naf, naf(T1, adv("provably")), n, forall, N, P1, P3).
rule(neg, T1, n, n, N, P1, P2),
rule(adv, adv("provably"), n, n, P2, P3).

We still have to deal with the verb phrase (vp) on the sentence level that is part of rule (8). This verb phrase expands into a copula (cop), followed by an adjective (adj):

17. rule(vp, vp(T1, T2), Y, n, N, P1, P3).
rule(cop, T1, Y, n, N, P1, P2),
rule(adj, T2, n, n, N, P2, P3).

In ASP, these grammar rules are processed bottom-up and during this model generation process the following syntax tree is produced for our example sentence:

18. s(np(det("Every"),
nl(noun("student"),
rcl(rcl(rp("who"),
  vp(iv("works"))),
cnj("and"),
rcl(rp("who"),
  vp(cop("is"),
    naf(neg("not"),
      adv("provably"),
      adj("absent")))))
  ,
vp(cop("is"),
adj("successful")),
  pm(".")))
This syntax tree needs to be translated into a suitable ASP representation for automated reasoning as we will see in Section 4. Before we do this, please note that it is relatively straightforward to generate look-ahead information that informs the author about the admissible input. To generate look-ahead information, we add one or more dummy tokens that contain a special string ($lah$) to the last input token, for example:

19. `token("Every", 1, 1, 2).
token("student", 1, 2, 3).
token("$lah$", 1, 3, 4).

Additionally, we add for each category a lexical entry that contains this special string, for example:

20. `lexicon(iv, "$lah$", n, sg, n).

The following ASP rules in (21) are then used to collect those syntax tree fragments that span the input string and contain look-ahead information. The last rule in (21) is a constraint and makes sure that the entire input string is considered:

21. `lah(C, T, Y, M, N, 1, P2) :-
    rule(C, T, Y, M, N, 1, P2),
    end_pos(P2, N).

lah :-
    lah(C, T, Y, M, N, 1, P2),
    end_pos(P2, N).

-end_pos(P2, N1) :-
    token("$lah$", N1, P1, P2),
    token("$lah$", N2, P3, P4),
    N1 <= N2, P2 < P4.

end_pos(P2, N) :-
    token("$lah$", N, P1, P2),
    not -end_pos(P2, N).

:- not lah.

In our example, the addition of the first token (`token("$lah$", 1, 3, 4)`) in (19) results in an unsatisfiable program. The addition of a further token (`token("$lah$", 1, 4, 5)`) results in two tree fragments that span the input string. From these trees, we can extract the relevant look-ahead information.

4 From Syntax Trees to Reified ASP Rules

We choose an indirect encoding for ASP rules where rules are reified as facts. This kind of encoding is necessary since there exists no mechanism within ASP that would allow us to assert new rules. The reified rules are then further processed
by an ASP meta-interpreter as we will see in Section 5. A reified rule consists of up to four different fact types: a fact type \( \text{rule}/1 \) for the identification of the rule, a fact type \( \text{head}/2 \) for the head of the rule, a fact type \( \text{pbl}/2 \) for positive body literals (if any), and a fact type \( \text{nbl}/2 \) for negative body literals (if any). For example, the translation of the syntax tree in (18) will result in the following encoding:

22. rule(1).
   head(1, lit(func(successful), arg(sk(1))))).
   pbl(1, lit(func(work), arg(sk(1))))).
   pbl(1, lit(func(student), arg(sk(1))))).
   nbl(1, lit(func(absent), arg(sk(1))))).

The fact type \( \text{rule}/1 \) stores the rule number \( 1 \). This rule number occurs as first argument in the other fact types and specifies rule membership. The actual literals that belong to a rule are encoded with the help of the term \( \text{lit}/2 \) where the first argument (e.g., \( \text{func(successful)} \)) is the functor name of the literal and the second argument \( \text{arg(sk(1))} \) represents a Skolem constant \( 1 \) that replaces the variable in the literal. Note that all facts that have been derived from rules need to be grounded and cannot contain any variables.

In the following, we will show in detail how the syntax tree in (18) is translated into the proposed ASP notation for rules in (22). The syntax tree is first split into three main parts: a part \( \text{to\_qnt}/3 \) to be translated into a quantifier, a part \( \text{to\_body}/3 \) to be translated into a rule body, and a part \( \text{to\_head}/3 \) to be translated into a rule head. In our case, these parts correspond to the determiner (det), the nominal expression (n1), and the verb phrase (vp) of the first sentence in (5):

23. to\_qnt(N, det("Every"), M) :-
   rule(det, det("Every"), Y, M, N, P1, P2),
   rule(n1, T2, Y, M, N, P2, P3),
   rule(vp, T3, Y, n, N, P3, P4).

   to\_body(N, T2, M) :-
   rule(det, det("Every"), Y, M, N, P1, P2),
   rule(n1, T2, Y, M, N, P2, P3),
   rule(vp, T3, Y, n, N, P3, P4).

   to\_head(N, T3, M) :-
   rule(det, det("Every"), Y, M, N, P1, P2),
   rule(n1, T2, Y, M, N, P2, P3),
   rule(vp, T3, Y, n, N, P3, P4).

In the next step, the determiner ("Every") is processed and this results in a new predicate \( \text{qnt}/4 \) that stores a rule number (\( R \)), the universal quantifier (\( M \)), the sentence number (\( N \)), and a Skolem constant (\( K \)) for the universal quantifier. Note that the rule number and the number for the Skolem constant are generated with the help of Lua\(^1\) and assigned (=) to the variables (\( R \)) and (\( K \)):

\(^1\) The number \( i \) in \( \text{sk}(i) \), represents the \( i \)th Skolem constant.

\(^2\) Lua [http://www.lua.org] is available as integrated scripting language in clingo.
24. qnt(R, M, N, sk(K)) :-
   to_qnt(N, det("Every"), M),
   R := @rule_num(),
   K := @sk_num().

Given the new predicate (qnt/4) for the universal quantifier, the syntax tree fragments for constructing the head of a rule and the body of a rule can be further split up:

25. to_head(R, adj(S2), K) :-
   to_head(N, vp(cop(S1), adj(S2)), M),
   qnt(R, forall, N, K).

to_body(R, noun(S), K) :-
   to_body(N, n1(noun(S), RCL), M),
   qnt(R, forall, N, K).

to_body(R, RCL, K) :-
   to_body(N, n1(noun(S), RCL), M),
   qnt(R, forall, N, K).

In the case of the head (to_head/3), this process results in a preterminal category (adj(S2)) that can be used to generate the head literal of the rule. In the case of the body (to_body/3), only the first rule generates a preterminal category (noun(S)) that can directly be used to generate a positive body literal. The second rule is used to split the relative clause (RCL) into its basic constituents in order to extract the relevant preterminal categories:

26. to_body(R, RCL1, K) :-
   to_body(R, rcl(RCL1, cnj(and), RCL2), K).

to_body(R, RCL2, K) :-
   to_body(R, rcl(RCL1, cnj(and), RCL2), K).

to_body(R, iv(S2), K) :-
   to_body(R, rcl(rp(S1), vp(iv(S2))), K).

to_body(R, naf(adj(S3)), K) :-
   to_body(R, rcl(rp(S1), vp(cop(S2), naf(T1, T2), adj(S3))), K).

The preterminal categories for content words together with the Skolem constant (K) and the rule number (R) are then used to generate the head literal, the positive and negative body literals. During this process the string (S) of these preterminal categories is replaced by the base form (B) via a lexicon lookup. The rule identifier (rule/1) is generated with the help of the head literal (head/2):

27. rule(R) :- head(R, L).

head(R, lit(func(B), arg(K))) :-
   to_head(R, adj(S), K),
   lexicon(adj, S, B, _, _).
pbl(R, lit(func(B), arg(K))) :-
    to_body(R, noun(S), K),
    lexicon(noun, S, B, _, _).

pbl(R, lit(func(B), arg(K))) :-
    to_body(R, iv(S), K),
    lexicon(iv, S, B, _, _).

nbl(R, lit(func(B), arg(K))) :-
    to_body(R, naf(adj(S)), K),
    lexicon(adj, S, B, _, _).

Note that checking for anaphoric references can be done over the existing
model during the translation process of the syntax tree into rules. For example,
the second sentence of (5) contains a definite noun phrase \textit{(the student)}
that is anaphorically linked to an indefinite noun phrase \textit{(a student)}. Depending on the
context in which an anaphoric expression occurs, we either check the body of
the current rule for an antecedent or the heads of all existing rules that don’t
have a body and give preference to the closest match in terms of rule numbers.

5 Reasoning with Reified ASP Rules

In order to process these reified ASP rules, we use a meta-interpreter that is
based on the work of Eiter et al. \cite{eiter2006}. We substantially extended this meta-
interpreter so that it can deal with variables that occur as Skolem constants
in the reified notation. On the meta-level we represent answer sets with the help
of the predicate \textit{in\_AS/1} and use the following two rules to add literals to an
answer set:

28. \textit{in\_AS}(lit(F, A2)) :-
    head(R, lit(F, A1)),
    pos_body_true(R, A1, A2),
    not neg_body_false(R, A1, A2).

\textit{in\_AS}(lit(F, A)) :-
    head(R, lit(F, A)),
    rule(R),
    not pos_body_exists(R).

The first rule specifies that a literal \textit{(lit/2)} is in an answer set \textit{(in\_AS/1)},
if it occurs in the head \textit{(head/2)} of a rule with number \textit{R} whose positive body
\textit{(pos\_body\_true/3)} is true and whose negative body \textit{(neg\_body\_false/3)} is not false
and if the Skolem constant that occurs as argument \textit{(A1)} of that literal can be
replaced by other constants that occur as argument \textit{(A2)} of a corresponding
literal in the answer set. The second rule specifies that if no positive body literal
\textit{(pbl/3)} for a rule exists, then we can directly process the head \textit{(head/2)} of a rule.

The positive body \textit{(pos\_body\_true/3)} of the first rule in (28) is true up to
some positive body literal with respect to a built-in order. If the positive body
is true up to the last positive body literal then the whole positive body is true. The first rule in (29) deals with this case; the second rule takes care of the first positive body literal, and the third rule makes sure that the positive body literals follow the specified order:

29. `pos_body_true(R, A1, A2) :-
    pos_body_true_up_to(R, F, A1, A2),
    not pbl_not_last(R, F).`

29. `pos_body_true_up_to(R, F, A1, A2) :-
    pbl_in_AS(R, F, A1, A2),
    not pbl_not_first(R, F).`

29. `pos_body_true_up_to(R, F1, A1, A2) :-
    pbl_in_AS(R, F1, A1, A2),
    F2 < F1,
    not pbl_in_between(R, F2, F1),
    pos_body_true_up_to(R, F2, A1, A2).`

The rule (`pbl_in_AS/4`) in (30) checks if a positive body literal (`pbl/2`) for a rule (`R`) exists, looks in the current answer set (`in_AS/1`) for a literal that has the same functor name (`F`) as the body literal but shows a different argument (`A2`) and returns that argument:

30. `pbl_in_AS(R, F, A1, A2) :-
    pbl(R, lit(F, A1)),
    in_AS(lit(F, A2)),
    A1 != A2.`

There exist similar rules that deal with cases where the positive body literal has more than one argument. The successor relation on positive body literals of each rule is defined with the help of the following auxiliary rules:

31. `pbl_in_between(R, F1, F3) :-
    pbl(R, lit(F1, A1)),
    pbl(R, lit(F2, A2)),
    pbl(R, lit(F3, A3)),
    F1 < F2, F2 < F3.`

31. `pbl_not_last(R, F1) :-
    pbl(R, lit(F1, A1)),
    pbl(R, lit(F2, A2)),
    F1 < F2.`

31. `pbl_not_first(R, F1) :-
    pbl(R, lit(F1, A1)),
    pbl(R, lit(F2, A2)),
    F2 < F1.`

The negative part of the body (`neg_body_false/3`) in the first rule of (28) is false, if it can be shown that one of its literals is in the answer set (`in_AS/1`). The rule in (32) checks this condition for literals with one argument (other rules deal with literals that have more than one argument):
32. neg_body_false(R, A1, A2) :-
    nbl(R, lit(F, A1)),
    in_AS(lit(F, A2)).

Finally, the rule pos_body_exists/1 in (33) is used as part of the second rule in (28) and simply checks if a positive body literal (pbl/2) exists:

33. pos_body_exists(R) :- pbl(R, L).

After parsing and translating the CNL text in (5) into reified rules represented as a set of facts, the ASP meta-interpreter will generate the following answer set as solution:

34. { in_AS(lit(func(student), arg(john)))
    in_AS(lit(func(work), arg(john)))
    in_AS(lit(func(student), arg(sue)))
    in_AS(lit(func(work), arg(sue)))
    in_AS(lit(func(student), arg(mary_ann)))
    in_AS(lit(func(absent), arg(mary_ann)))
    in_AS(lit(func(successful), arg(sue)))
    in_AS(lit(func(successful), arg(john)))
    in_AS(lit(func(neg(work)), arg(mary_ann)))
}

This answer set contains the same information as the answer set in (3) and can be used for question answering.

6 Conclusion

In this paper, we investigated in an experimental way if it is possible to process a controlled natural language entirely in ASP and if ASP can serve as a unified framework for parsing, knowledge representation and automated reasoning. ASP is a powerful declarative knowledge representation language that provides support for non-monotonic reasoning and this makes the language particularly attractive for controlled natural language processing. We showed in detail how a grammar for a controlled natural language can be written as an ASP program. This grammar is processed bottom-up and the syntax trees are constructed starting from the leaves up to the root. The resulting syntax trees are translated into reified rules that consist of a set of facts. These facts are then used by a meta-interpreter written in ASP for automated reasoning. The translation into reified rules is necessary because ASP does not provide a mechanism that would allow us to generate and assert normal ASP rules in the same program. Alternatively, we could take the resulting syntax trees and translate them outside of the ASP program into normal ASP rules and then generate a new ASP program that executes these rules. With the help of the presented ASP grammar it is possible to generate look-ahead information to guide the writing process of the author. It is also possible in ASP to perform anaphora resolution over the reified rules during the translation process using the standard constraints on anaphoric accessibility. We believe that ASP is an interesting paradigm for controlled natural language processing and plan to extend the presented approach or aspects of it and integrate them into a controlled language authoring system.
References

1. Brewka, G., Eiter, T. and Truszczynski M.: Answer Set Programming at a Glance. In: Communications of the ACM, Vol. 54, No. 12, December (2011)
2. Clark, P., Harrison, P., Jenkins, T., Thompson, J. and Wojcik, R.H.: Acquiring and using world knowledge using a restricted subset of English. In: Proceedings of FLAIRS 2005, AAAI Press, pp. 506–511, (2005)
3. Eiter, T., Faber, W., Leone, N. and Pfeifer, G.: Computing Preferred Answer Sets by Meta-Interpretation in Answer Set Programming. in: INFSYS Research Report 1843-02-01, Technische Universit¨ at Wien, January, (2002)
4. Franconi, E., Guagliardo, P., Trevisan, M. and Tessaris S.: Quelo: an ontology-driven query interface. In: Proceedings of the 24th International Workshop on Description Logics (DL 2011), (2011)
5. Fuchs, N.E., Kaljurand, K. and Kuhn, T.: Attempto Controlled English for knowledge representation. In: Reasoning Web – 4th International Summer School 2008, LNCS 5224, Springer, pp. 104–124, (2008)
6. Gebser, M., Kaminski, R., Kaufmann, B., Ostrowski, M., Schaub, T. and Schneider, M.: Potassco: The Potsdam Answer Set Solving Collection. In: AI Communications, Vol. 24, No. 2, pp. 105–124, (2011)
7. Gelfond, M. and Lifschitz V.: The stable model semantics for logic programming. In: Proceedings of ICLP-88, pp. 1070–1080, (1988)
8. Gelfond, M. and Lifschitz V.: Classical negation in logic programs and disjunctive databases. In: New Generation Computing, Vol. 9, No. 34, pp. 365–386, (1991)
9. Gelfond, M. and Kahl, Y.: Knowledge Representation, Reasoning, and the Design of Intelligent Agents. Cambridge University Press, (2014)
10. Kuhn, T.: Controlled English for Knowledge Representation. Doctoral thesis, Faculty of Economics, Business Administration and Information Technology of the University of Zurich, (2010)
11. Kuhn, T.: A Survey and Classification of Controlled Natural Languages. In: Computational Linguistics, Vol. 40, No. 1, pp. 121–170, (2014)
12. Lierler, Y. and Lifschitz, V.: Logic Programs vs. First-Order Formulas in Textual Inference. In: Proceedings of the 10th International Conference on Computational Semantics (IWCS 2013), Potsdam, Germany, pp. 340–346, (2013)
13. Lifschitz, V.: What is Answer Set Programming? In: Proceedings of AAAI 2008, pp. 1594–1597, (2008)
14. Power R.: OWL Simplified English: a finite-state language for ontology editing. In: Kuhn, T. and Fuchs, N.E. (eds): Proceedings of CNL 2012, Zurich, Springer, Heidelberg, pp. 44–60, (2012)
15. Reiter, R.: On closed world data bases. In: Gallaire, H. and J. Minker, J. (eds): Logic and Data Bases, pp. 119–140, Plenum Publ. Co., New York, (1978)
16. Schwitter, R., Ljungberg, A. and Hood, D.: ECOLE – A Look-ahead Editor for a Controlled Language. In: Proceedings of EAMT-CLAW03, May 15-17, Dublin City University, Ireland, pp. 141–150, (2003)
17. Schwitter, R.: Controlled Natural Languages for Knowledge Representation. In: Proceedings of COLING 2010, Beijing, China, pp. 1113–1211, (2010)
18. White, C. and Schwitter, R.: An Update on PENG Light. In: L. Pizzato and R. Schwitter (eds.), Proceedings of ALTA 2009, Sydney, Australia, pp. 80–88, (2009)